/home/m262f99/A=B/Maple/koepf4.7.mws

Math 262a, Fall 1999, Glenn Tesler
Showing sums are equal, without being able to evaluate the sums in closed form
10/25/99

Koepf # 4.7

For nonnegative integer n, prove

```
> Sum(binomial(n,k)^3,k=0..n)=Sum(binomial(n,k)^2*binomial
  (2*k,n),k=0..n);
```

$$\sum_{k=0}^{n} \mathrm{binomial}(n,k)^3 = \sum_{k=0}^{n} \mathrm{binomial}(n,k)^2 \,\mathrm{binomial}(2\,k,n)$$

The bounds on both sums are the natural bounds, so we may sum k=-infinity..infinity.
We will show that both sums satisfy the same recursion and initial conditions.
(It is NOT necessary that the algorithm discover the same recursion for both sums, even if they are equal; later we'll learn about noncommutative LCM's and GCD's and the Euclidean algorithm for dealing with that.)

```
> read EKHAD;
```

Version of Feb 25, 1999

(verbiage deleted)

```
> F1 := (n,k)->binomial(n,k)^3;
  zeilpap(F1(n,k),k,n);
```

$$F1 := (n,k) \to \mathrm{binomial}(n,k)^3$$

```
A PROOF OF A RECURRENCE

By Shalosh B. Ekhad, Temple University, ekhad@math.temple.edu

Theorem:Let   F(n,k)   be given by
```

$$\mathrm{binomial}(n,k)^3$$

```
and let   SUM(n)   be the sum of   F(n,k)    with
respect to   k   .

SUM(n)    satisfies the following linear recurrence equation
```

$$-8\,(n+1)^2\,\mathrm{SUM}(n) + (-7\,n^2 - 21\,n - 16)\,\mathrm{SUM}(n+1) + (n+2)^2\,\mathrm{SUM}(n+2) = 0.$$

PROOF: We cleverly construct G(n,k) :=

$$k^3 (-72 + 78 k - 272 n - 290 n^3 - 402 n^2 - 102 n^4 - 14 n^5 + 147 n^3 k + 27 n^4 k - 66 n^2 k^2 - 18 n^3 k^2 + 4 n^2 k^3 + 8 n k^3 + 4 k^3 + 291 n^2 k - 30 k^2 + 249 n k - 78 n k^2) \operatorname{binomial}(n, k)^3 \Big/ ((-n - 1 + k)^3 (-n - 2 + k)^3)$$

with the motive that

$$-8 (n + 1)^2 \mathrm{F}(n, k) + (-7 n^2 - 21 n - 16) \mathrm{F}(n + 1, k) + (n + 2)^2 \mathrm{F}(n + 2, k)$$

= G(n,k+1)-G(n,k) (check!)

and the theorem follows upon summing with respect to k .QED.

```
> F2 := (n,k) -> binomial(n,k)^2 * binomial(2*k,n);
  zeilpap(F2(n,k),k,n);
```

$$F2 := (n, k) \to \operatorname{binomial}(n, k)^2 \operatorname{binomial}(2 k, n)$$

A PROOF OF A RECURRENCE

By Shalosh B. Ekhad, Temple University, ekhad@math.temple.edu

Theorem:Let F(n,k) be given by

$$\operatorname{binomial}(n, k)^2 \operatorname{binomial}(2 k, n)$$

and let SUM(n) be the sum of F(n,k) with respect to k .

SUM(n) satisfies the following linear recurrence equation

$$\mathbf{-8 (n + 1)^2 SUM(n) + (-7 n^2 - 21 n - 16) SUM(n + 1) + (n + 2)^2 SUM(n + 2)}$$

$$= 0.$$

PROOF: We cleverly construct G(n,k) :=

$$\frac{(n + 1) k^2 (-3 n - 6 + 2 k) (2 k - n - 1) (2 k - n) \operatorname{binomial}(n, k)^2 \operatorname{binomial}(2 k, n)}{(-n - 1 + k)^2 (-n - 2 + k)^2}$$

with the motive that

$$-8 (n + 1)^2 \mathrm{F}(n, k) + (-7 n^2 - 21 n - 16) \mathrm{F}(n + 1, k) + (n + 2)^2 \mathrm{F}(n + 2, k)$$

= G(n,k+1)-G(n,k) (check!)

and the theorem follows upon summing with respect to k .QED.

Notice both sums satisfy the same recursion of order 2! The top order term has a leading zero coefficient if n=-2, which is not a nonnegative integer, so it’s not of concern. Thus, as long as both sums agree for n=0 and n=1 (i.e., the first 2 values), iterating the recursion will make all future values equal as well. Here we check f1(0),...,f1(10) to demonstrate this (but to prove it, only f1(0), f1(1) are needed):

```
> f1 := n -> expand(sum(F1(n,k),k=0..n));
  'f1(nn)'$'nn'=0..10;
```

$$f1 := n \rightarrow \operatorname{expand}\left(\sum_{k=0}^{n} \mathrm{F1}(n, k)\right)$$

$$1, 2, 10, 56, 346, 2252, 15184, 104960, 739162, 5280932, 38165260$$

And the same for f2:

```
> f2 := n -> expand(sum(F2(n,k),k=0..n));
  'f2(nn)'$'nn'=0..10;
```

$$f2 := n \rightarrow \operatorname{expand}\left(\sum_{k=0}^{n} \mathrm{F2}(n, k)\right)$$

$$1, 2, 10, 56, 346, 2252, 15184, 104960, 739162, 5280932, 38165260$$

And we can have them compared directly, too:

```
> 'expand(f1(nn)-f2(nn))'$nn=0..10;
```

$$0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0$$

```
>
```